The Integral Representation of the Bessel Functions of the First Kind

Let's give the

$$\int_0^{\text{Pi}} \cos(x\sin(t) - n\,t)\,dt$$

integral values with the help of the functions of Maple in which case the n=0,1,2,... can be a natural number and the x can be an arbitrary real number

It is important to know that the first edition of our book was made in a previous version of Maple where we devoted a whole worksheet to the calculation of the $\int_0^{\frac{\pi}{2}} \sin(x)^n\,dx$ paraméteres integrál kiszámítására. parametric integral. Let's look what the response of the system would be nowadays.

> $\int_0^{\frac{1}{2}\text{Pi}} \sin(x)^n\,dx$

$$\frac{1}{2}\,\frac{\sqrt{\pi}\,\Gamma\left(\frac{1}{2}+\frac{1}{2}\,n\right)}{\Gamma\left(1+\frac{1}{2}\,n\right)} \qquad (1)$$

Well, times have changed and with the course of time, our previous worksheet has also changed. Nowadays Maple can calculate the integral in one step so we had to look for another task the solution of which was nearly as ingenious as the solution of the original integral.

> *restart*

The values of the integrals given in the task depend on the two variables, the n and the x. Thus we enter the integrals as functions with two variables. The n can be an integer that's why we assume the same in the case of their creation with the help of the assuming instruction.

> $J := (n, x) \rightarrow \int_0^{\text{Pi}} \cos(x\sin(t) - n\,t)\,dt$

$$J := (n, x) \rightarrow \int_0^{\pi} \cos(x\sin(t) - n\,t)\,dt \qquad (2)$$

> *assuming*([('J')(n, x) = J(n, x)], [n:: *posint*])

$$J(n, x) = \int_0^{\pi} \cos(-x\sin(t) + n\,t)\,dt \qquad (3)$$

The result is unfortunate. After a short consideration, the system returned the command given. It usually happens like this if the system is unable to execute the command given. In our case this means that the system did not find a closed formula for the integral we had been looking for.

What can be done in this situation? We have to think it over and execute it more carefully. Instead of the one-step solution of the task, let's calculate the J(n,x) integrals for concrete n values. Maybe this will lead to success.

> *simplify*([*seq*(*J*(*n*, *x*), *n* = 0 ..3)], *symbolic*);

$$\left[\pi\,\mathrm{BesselJ}(0, x), \int_0^{\pi} \cos(x \sin(t) - t)\,\mathrm{d}t, \int_0^{\pi} \cos(x \sin(t) - 2\,t)\,\mathrm{d}t, \int_0^{\pi} \cos(x \sin(t) - 3\,t)\,\mathrm{d}t\right] \qquad (4)$$

Great! Although the system could not cope with the integrals for the n=1,2,3 it offered a glimmer of hope concerning the solution because it recognised the n=, namely, that J(0,x)=PiBesselJ(0,x). So according to the n=0, the solution of the task should be found in the field of the BesselJ functions. Let's ask the system about the BesselJ function itself

> *help*("BesselJ")

The help(BesselJ); command opens a new window on the screen where we can find some information about the Bessel functions written in English.



As we can see, we were told that each of the BesselJ(n,x) and BesselY(n,x) function was a solution to the

$$x^2\ y''(x) + x\ y'(x) + (x^2 - n^2)\ y(x) = 0$$

common differential equation. Let's check it by searching for the general solution. For this, create the differential equation then solve it with the dsolve procedure.

> $x^2\ y''(x) + x\ y'(x) + (x^2 - n^2)\ y(x) = 0$

$$x^2 \left(\frac{\mathrm{d}^2}{\mathrm{d}x^2}\, y(x)\right) + x \left(\frac{\mathrm{d}}{\mathrm{d}x}\, y(x)\right) + (x^2 - n^2)\, y(x) = 0 \qquad (5)$$

> *dsolve*(**??**, $y(x)$)

$$y(x) = \_C1 \operatorname{BesselJ}(n, x) + \_C2 \operatorname{BesselY}(n, x) \quad \textbf{(6)}$$

It looks fine so far. We are told that the ?? differential equation has two independent solutions. But we are interested in the J(n,x) integrals thus it crosses our mind that the solutions of the differential equations can be created with the uselnt option. Let's try this as well.

> *dsolve*(**??**, $y(x)$, *useInt*)

$$y(x) = \_C1 \operatorname{BesselJ}(n, x) + \_C2 \operatorname{BesselY}(n, x) \quad \textbf{(7)}$$

It has not returned a result. So far our efforts have failed. Let's try the graphic tools. Maple can numerically evaluate the J(n,x) integrals. This enables the representation of the J(n,x) functions. Let's draw the curves of the $J(0, x), J(1, x), J(2, x), J(3, x)$ functions.

> *plot*(**(4)**, $x = 0 ..10$)



This looks good: as if we were looking at the graphs of a vibration with damping. For the sake of comparison, draw the BesselJ(n, x) BesselY(n, x)unctions for the n=0,1,2,3 as well.

> *plot*([BesselJ$(0, x)$, BesselJ$(1, x)$, BesselJ$(2, x)$, BesselJ$(3, x)$], $x = 0 ..10$)



No doubt. The curves in the two graphs are identical concerning their kind. Moreover, the difference between the scaling of the y axes perfectly illustrates the $J(0, x) =$ Pi BesselJ$(0, x)$ equality known. After this, let's draw the curves of the BesselY functions for the fun of it.

> *plot*([BesselY(0, *x*), BesselY(1, *x*), BesselY(2, *x*), BesselY(3, *x*)], *x* = 0 ..10, *y* = −10 ..1)



We assume that we have gained enough experience to come up with our conjecture according to which the J(n,x) integrals appeared in the task are the πth of the BesselJ functions, that is,

$$J(n,x) = \int_0^{Pi} \cos(x \sin(t) - n\,t)\, dt = \text{Pi BesselJ}(n,x)\,, \quad (n=0, 1, 2, ...\ ).$$

Only the proof is ahead.

According to the existence and uniqueness theorems of the differential equations, the

$$a(x)\, y''(x) + b(x)\, y'(x) + c(x)\, y(x) = 0$$

second-order linear homogenous common differential equation

$$y(0) = y_0 \text{ és } y'(x)\Big|_{x=0} = y_1,$$

has only one solution with an arbitrary initial assumption if the ax, b(x) and c(x) are continuous functions. Notice that the ?? differential equation satisfies these assumptions. According to this, the uniqueness of the solutions of the ?? is ensured. Since we know that the BesselJ(n,x) is the solution to the ??, by showing that the $J(n,x) = \pi.BesselJ(n,x)$ function is also a solution to the ?? with the same initial assumptions the $\frac{J(n,x)}{\pi}$ equation can be proved.

First, let's show that the J(n,x) integrals satisfy the ?? differential equations just like the BesselJ(n,x) functions. Notice that it is sufficient because in this case their $\frac{1}{\pi}$ fold is also going to be the solution to the differential equation.

Look at the first and second derivatives of the J(n,x) functions. Then substitute into the ?? differential equation.

> $\frac{d}{dx} J(n, x)$

$$\int_0^{\pi} \sin(-x \sin(t) + n t) \sin(t) \, dt \quad \textbf{(8)}$$

> $\frac{d^2}{dx^2} J(n, x)$

$$\int_0^{\pi} -\cos(-x \sin(t) + n t) \sin(t)^2 \, dt \quad \textbf{(9)}$$

> $subs\left(\left[y(x) = J(n, x), \frac{d}{dx} y(x) = \int_0^{Pi} \sin(-x \sin(t) + n t) \sin(t) \, dt, \frac{d^2}{dx^2} y(x) = \int_0^{Pi} -\cos(-x \sin(t) + n t) \sin(t)^2 \, dt\right], \textbf{(5)}\right)$

$$x^2 \int_0^{\pi} -\cos(-x \sin(t) + n t) \sin(t)^2 \, dt + x \int_0^{\pi} \sin(-x \sin(t) + n t) \sin(t) \, dt + (x^2 - n^2) \int_0^{\pi} \cos(-x \sin(t) + n t) \, dt = 0 \quad \textbf{(10)}$$

Combine the additivity of the integrals with the identities of the functions.

> $combine(\%)$

$$x \sin(n \pi) - n \sin(n \pi) = 0 \quad \textbf{(11)}$$

The built-in identities of Maple operate perfectly, just as we expected. The left side will not be zero for all the ns only if n is an integer because the sin(nPi) is zero for these ns. The simplify procedure recognizes this if we give that the n can be an integer in the assuming option.

> $assuming([simplify(\%)], [integer])$

$$0 = 0 \quad \textbf{(12)}$$

So the J(n,x) is really a solution to (5) in the case of all 0<n integers. Now we have to check the initial assumptions only in the case of 0<n because Maple gave the answer for the =0 when we got the $J(0, x) = \text{Pi BesselJ}(0, x)$ equality in the (4) output.

Let's start with the evaluation of the integrals of the task considered at the x=0 with the 0<n assumption.

> $('J')(n, 0) = J(n, 0)$

(13)

$$J(n, 0) = \frac{\sin(n\pi)}{n} \quad \textbf{(13)}$$

> *assuming*([('J')(n, 0) = *rhs*(%)], [n:: *posint*]);

$$J(n, 0) = 0 \quad \textbf{(14)}$$

Continue with the evaluation of the BesselJ(n,0).

> *assuming*([*simplify*(BesselJ(n, 0))], [n:: *posint*])

$$\text{BesselJ}(n, 0) \quad \textbf{(15)}$$

This is bad news for us. In the case of an arbitrary positive n integer the system cannot give the BesselJ (n,0) values, although all of them should be zero. The calculation of the BesselJ(n,0) works for concrete n values but unfortunately the proof cannot be represented for all the ns.

> *seq*(BesselJ(k, 0), k = 0..10)

$$1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \quad \textbf{(16)}$$

What can be done in this case? Maple can help us with the FunctionAdvisor procedure which gives some definition advice about the built-in functions. Let's try what the FunctionAdvisor procedure knows about the BesselJ functions.

> *FunctionAdvisor*(*definition*, BesselJ(n, x))

$$\left[\text{BesselJ}(n, x) = \frac{x^n \,\text{hypergeom}\left([\,], [1 + n], -\frac{1}{4} x^2\right)}{\Gamma(1 + n)\, 2^n}, \textit{with no restrictions on } (n, x)\right] \quad \textbf{(17)}$$

AThe answer is that the BesselJ(n,x) functions can be given with the help of the hypergeometric functions in which case no restrictions have to be given for the n and x. Now let's continue with the next question: what can be the definition of the hypergeometric functions?

> *FunctionAdvisor*(*definition*, hypergeom([ ], [j], x))

$$\left[\text{hypergeom}([\,], [j], x) = \sum_{\_k1 = 0}^{\infty} \frac{x^{\_k1}}{\_k1!\, \text{pochhammer}(j, \_k1)}, \textit{with no restrictions on } (j, x)\right] \quad \textbf{(18)}$$

The hypergeometric function is a power series in the x. But we have not finished with the unknown references because there is something that should be known about the pochhammer functions.

> *FunctionAdvisor*(*definition*, pochhammer(j, k))

$$\left[\text{pochhammer}(j, k) = \frac{\Gamma(j + k)}{\Gamma(j)}, \text{And}(j::(\text{Not}(\textit{nonposint})), (j + k)::(\text{Not}(\textit{nonposint})))\right] \quad \textbf{(19)}$$

Euler's GAMMA function appeared at the beginning of the worksheet. This is the generalisation of the factorial values for the non-integer numbers. The convert procedure of Maple converts the GAMMA (k+1) value to k!..

> GAMMA(k + 1) = *convert*(GAMMA(k + 1), factorial);

$$\Gamma(k + 1) = k! \quad \textbf{(20)}$$

It is time for us to assemble the definitions embedded thus we can get the power series of the BesselJ(n, x) like a puzzle from its elements.
According to the first definition, the sum of the hypergeometric row has to be multiplied by the

$\dfrac{\left(\frac{x}{2}\right)^n}{n!}$ factor. The hypergeometric row goes according to the powers of $\left(-\frac{x^2}{4}\right)$ in the denominator of which there are the product of the k! and the $\dfrac{(n+k)!}{n!}$ t factors. Notice that we can simplify with the n! in the elements of the row thus getting the following formula for the syntax of the BesselJ(n,x) power series.

> $hatványsor := \left(\frac{x}{2}\right)^n \left(\sum_{k=0}^{infinity} \frac{\left(-\frac{x^2}{4}\right)^k}{k!\ (n+k)!}\right)$

$$hatványsor := \left(\frac{1}{2}\,x\right)^n \left(\sum_{k=0}^{\infty} \frac{\left(-\frac{1}{4}\,x^2\right)^k}{k!\ (n+k)!}\right) \quad \textbf{(21)}$$

It would be great if Maple responded the BesselJ(n,x) as the sum of the row.

> $value(\%);$

$$\frac{\left(\frac{1}{2}\,x\right)^n \mathrm{BesselJ}(n,x)\,\Gamma(n+1)\,2^n}{n!\,x^n} \quad \textbf{(22)}$$

> $simplify(\%)$

$$\mathrm{BesselJ}(n,x) \quad \textbf{(23)}$$

Great! We have succeeded. By using the calculation of the row we can easily finish the proof of the [képlet] initial assumptions for every positive n because this can be simply deducted from the fact that the $\left(\frac{x}{2}\right)^n$ factor located before the infinite sum is zero in the case of x=0 and 0<n.

> $\left.\left(\left(\frac{x}{2}\right)^n \left(\sum_{k=0}^{infinity} \frac{\left(-\frac{x^2}{4}\right)^k}{k!\ (n+k)!}\right)\right)\right|_{x=0}$ $\quad assuming\ n :: posint$

$$0 \quad \textbf{(24)}$$

We also have to check the equalities concerning the initial assumptions of the differentiated functions in the case of all 0<n; integers.

$$\left.\left(\frac{d}{dx} J(n,x)\right)\right|_{x=0} = \mathrm{Pi}\left(\left.\left(\frac{d}{dx}\mathrm{BesselJ}(n,x)\right)\right|_{x=0}\right)$$

Let's start its proof with the evaluation of the derivates by the x of the integrals considered at the x=0.

> $\left.\left(\frac{d}{dx} J(n,x)\right)\right|_{x=0}$

$$-\frac{\sin(n\,\pi)}{-1+n^2} \quad \textbf{(25)}$$

With the exception of the n=1, in which case the denominator would be zero, the value of the derivative considered at x=0 is 0.

> $assuming([simplify(\%)], [and(n > 1, n :: posint)])$

$$0 \tag{26}$$

Let's calculate the n=1 exceptional case separately.

> $\left(\frac{d}{dx} J(1, x)\right)\Big|_{x=0}$

$$\frac{1}{2}\pi \tag{27}$$

According to this, the value considered at x=0 of every $\frac{d}{dx} J(n, x)$ derivative is zero, except for the n=1 case in which case we get $\frac{Pi}{2}$. So far it has been fine but we still have work to do. We have to show that the behaviour of the derivative of the BesselJ function is the same. More precisely that the values of the $\frac{d}{dx}$ BesselJ(n, x) derivatives are zero at the x=0, except for the n=1 in which case we expect to get the value of $\frac{1}{2}$. What is Maple going to give for the derivative by the x of the BesselJ(n,x) functions?

> $\frac{d}{dx} \text{BesselJ}(n, x)$

$$-\text{BesselJ}(1+n, x) + \frac{n\,\text{BesselJ}(n, x)}{x} \tag{28}$$

Although we have received a recurrence relation

$$\frac{\partial}{\partial x} \text{BesselJ}(n, x) = -\text{BesselJ}(1+n, x) + \frac{n\,\text{BesselJ}(n, x)}{x}$$

for the derivatives by the x of the BesselJ(n,x) functions, the beauty of the formula does not make up for the fact that it is useless for x=0. It is due to the x located in the denominator of the second term. So now we have to use the calculation of the row of the BesselJ(n,x) functions and derivate by the x again. It is known that the power series can be differentiated term by term inside the domain of the convergence. According to the advice of the FunctionAdvisor the power series of the BesselJ(n,x) function approaches in the case of an arbitrary x.

> $\frac{d}{dx}\, hatványsor$

$$\frac{\left(\frac{1}{2}x\right)^n n \left(\sum_{k=0}^{\infty} \frac{\left(-\frac{1}{4}x^2\right)^k}{k!\,(n+k)!}\right)}{x} + \left(\frac{1}{2}x\right)^n \left(\sum_{k=0}^{\infty} \frac{2\left(-\frac{1}{4}x^2\right)^k k}{x\,k!\,(n+k)!}\right) \tag{29}$$

> $deriváltsor := simplify(\%)$

$$deriváltsor := x^{(-1+n)} \left(2^{(-n)} n \left(\sum_{k=0}^{\infty} \frac{\left(-\frac{1}{4}x^2\right)^k}{\Gamma(k)\,k\,\Gamma(n+k+1)}\right) + 2^{(1-n)} \left(\sum_{k=0}^{\infty} \right. \right. \tag{30}$$

$$\frac{\left(-\frac{1}{4}x^2\right)^k}{\Gamma(k)\,\Gamma(n+k+1)}\Bigg)\Bigg)$$

This is getting more and more complicated. Is it going to end somehow?
Maple gave the derivation by considering the rules of the derivation of the function of the product. The simplify procedure highlighted the x^{n-1} power during the simplification and the other factor is the sum of the two power series. It is encouraging that in the case of 1<n we have the opportunity to calculate the substitution value considered at the x=0.

> *assuming*([*eval*(*deriváltsor*, *x* = 0)], [*and*(*n* > 1, *n* :: *posint*)])

$$0 \qquad \textbf{(31)}$$

We have expected this. We succeeded in proving that in the case of every n=2,3,4 integer the $\left(\frac{d}{dx}\,\mathrm{BesselJ}(n,x)\right)\Big|_{x=0} = 0$ eequality is satisfied.

Hold on because we have never been so close to the solution of the problem because only the n=1; case is ahead.
As we mentioned earlier, we expect $\frac{\mathrm{Pi}}{2}$ based on the $\left(\frac{d}{dx}\,\mathrm{BesselJ}(1,x)\right)\Big|_{x=0} = \frac{1}{2}$. value received in the case of the n=1 for the integers. To prove this, convert the differentiated row to one indefinite sum after the substitution of the n=1. Then consider the limit value of the row at the x=0.

> *eval*(*deriváltsor*, *n* = 1)

$$\frac{1}{2}\left(\sum_{k=0}^{\infty}\frac{\left(-\frac{1}{4}x^2\right)^k}{\Gamma(k)\,k\,\Gamma(2+k)}\right)+\sum_{k=0}^{\infty}\frac{\left(-\frac{1}{4}x^2\right)^k}{\Gamma(k)\,\Gamma(2+k)} \qquad \textbf{(32)}$$

> *simplify*(*combine*(%))

$$\sum_{k=0}^{\infty}\frac{1}{2}\,\frac{\left(-\frac{1}{4}x^2\right)^k(1+2k)}{\Gamma(k+1)^2\,(k+1)} \qquad \textbf{(33)}$$

> *limit*(%, *x* = 0)

$$\frac{1}{2} \qquad \textbf{(34)}$$

EWe have reached the end of the task and succeeded in proving that

$$\int_0^{\mathrm{Pi}}\cos(x\sin(t)-n\,t)\,dt = BesselJ(n,x) = \left(\frac{x}{2}\right)^n\left(\sum_{k=0}^{infinity}\frac{\left(-\frac{x^2}{4}\right)^k}{k!\,(n+k)!}\right)$$

is satisfied for every n=0,1,2,3,... integer by the combined usage of the differential equations of the BesselJ(n,x) functions and the calculations of the row.

What Have You Learnt About Maple?

• The int procedure of Maple is used to calculate the determined integral of the y=f(x) function considered in the [a,b] interval. Its call is int(f(x),x=a..b). The silent form of the int is the lnt procedure.

• During the conversion and the simplification of an expression, we can make assumptions with the assuming instructions by the syntax of

expression assuming assumption

• This is the syntax of the assuming in 1-D Math. It is a bit cumbersome that the syntax to be used in 2 -D Math is a bit different:

assuming([expression], [assumption])

The assumption is only valid in the given instruction.

• The z=f(x,y) function with two variables can be given in Maple with the $z := (x, y) \to f(x, y)$ assignment during which the two variables x and y of the function is put between brackets.

• The dsolve procedure gives the general solution of the common differential equation given in the first parameter by using known functions if it gets the name of the function being searched as its second parameter. So its syntax is:

dsolve(differential equation, x(x))

When using the dsolve procedure together with the uslnt option it tries to give the solution of the differential equation with the help of an integral. For example:

> $dsolve\left(diff(y(x), x) + \frac{(y(x))}{x} = 0, y(x), useInt\right)$

$$y(x) = \_C1\, e^{\left(\int -\frac{1}{x}\, dx\right)} \tag{35}$$

• The GAMMA function is among the built-in functions of Maple

$$\mathrm{GAMMA}(z) = \int_0^{infinity} e^{(-t)}\, t^{(z-1)}\, dt,$$

and the BesselJ function as well which is defined with the following convergent power series in the case of a positive n integer.

$$\mathrm{BesselJ}(n, x) = \left(\frac{x}{2}\right)^n \left(\sum_{k=0}^{infinity} \frac{(-1)^k \left(\frac{x}{2}\right)^{(2k)}}{k!\,(n+k)!}\right)$$

Exercises

1. Prove that in the case of k=0,1,2... positive integers it is GAMMA(k+1)=k! in which case the GAMMA function is defined by the following integral:

$$\text{GAMMA}(z) = \int_0^{\mathit{infinity}} e^{(-t)}\, t^{(z-1)}\, dt$$

2. By using the advice of the *FunctionAdvisor*(*definition*, *BesslY*(n, x)) instruction prove that if n is not an integer then BesselY(n,x) = $\dfrac{\text{BesselJ}(n, x)\cos(n\,\text{Pi}) - \text{BesselJ}(-n, x)}{\sin(n\,\text{Pi})}$

3. Give the values of the

$$\int_0^{\frac{\text{Pi}}{2}} \cos(x\cos(t))\sin(t)^{(2\,n)}\, dt$$

integrals with the help of the BesselJ(n,x) functions. (Notice the Poisson representation of the Bessel functions.)

4. Prove that the $z_1(x) = \sqrt{x}\ \text{BesselJ}(n, x)$ és a $z_2(x) = \sqrt{x}\ \text{BesselY}(n, x)$ functions are the solutions to the

$$\frac{d^2}{dx^2} z(x) + \left(1 + \frac{1 - 4\,n^2}{4\,x^2}\right) z(x) = 0$$

differential equation by using the dsolve procedure.

5. By using the result of task 4 prove that

$$z_1(x) \cdot \frac{d}{dx} z_2(x) - z_2(x) \cdot \frac{d}{dx} z_1(x) = \mathit{konstans}.$$

Determine the value of the constant.

6. By using the result of task 5 prove that the identity

$$\text{BesselJ}(n, x)\left(\frac{d}{dx}\text{BesselY}(n, x)\right) - \text{BesselY}(n, x)\left(\frac{d}{dx}\text{BesselJ}(n, x)\right) = \frac{2}{\text{Pi}\,x}$$

is true in the case of every 0<n integers.

7. Give the

$$\int_0^{\text{Pi}} \sin(x\sin(t) - n\,t)\, dt$$

integrals by using the functions of Maple in which case the n=0,1,2,... represents a natural number and the x represents a real number.